

## Системы аналитических вычислений

### Лабораторная работа 5

Арапов Степан М80-2086-19

#### Полиномы. НОД. Расширенный алгоритм Евклида.

1. Реализовать расширенный алгоритм Евклида.
2. Рассчитать НОД, сравнить его с результатом функции gcd().
3. Получить представление НОД в виде (тождество Безу).

По списку второй => вариант 2:

```
In[1]:= f = 5*x^4 - 15*x^3 + 15*x^2 - 5*x;  
g = -22*x^5 + 33*x^4 - 11*x^3 - 99*x + 99;  
f//TraditionalForm  
g//TraditionalForm
```

```
Out[3]//TraditionalForm=  
 $5x^4 - 15x^3 + 15x^2 - 5x$ 
```

```
Out[4]//TraditionalForm=  
 $-22x^5 + 33x^4 - 11x^3 - 99x + 99$ 
```

Какие корни у этих многочленов?

```
In[5]:= Roots[f == 0, x]  
Roots[g == 0, x, Reals]
```

```
Out[5]=  $x = 0 \vee x = 1 \vee x = 1 \vee x = 1$ 
```

**Roots** : Options expected (instead of  $\mathbb{R}$ ) beyond position 2 in Roots  $[99 - 99x - 11x^3 + 33x^4 - 22x^5 == 0, x, \mathbb{R}]$ . An option must be a rule or a list of rules .

```
Out[6]=  $x = 1 \vee x = \sqrt{-0.916 \dots - 1.02 \dots i} \vee x = \sqrt{-0.916 \dots + 1.02 \dots i} \vee$   
 $x = \sqrt{1.17 \dots - 1.02 \dots i} \vee x = \sqrt{1.17 \dots + 1.02 \dots i}$ 
```

Уже отсюда видно, что НОД f и g будет x-1.

Посмотрим на результат встроенной функции gcd.

```
In[7]:= gcdAuto = PolynomialGCD[f,g]
```

```
Out[7]=  $x - 1$ 
```

#### Расширенный алгоритм Евклида. Тождество Безу.

```

In[8]:= extEuclid[aa_, bb_] := Module[
  {a = aa, b = bb,
   u = 1, u1 = 0,
   v = 0, v1 = 1,
   quot, rem
  },

  While[Not[SameQ[b, 0]],
    quot = PolynomialQuotient[a, b, x]; (*возвращает частное от деления a и b*)
    rem = PolynomialRemainder[a, b, x]; (*возвращает остаток от деления a на b*)
    {a, b} = {b, rem};
    {u, u1} = {u1, (u - u1*quot)//ExpandAll};
    {v, v1} = {v1, (v - v1*quot)//ExpandAll};

  ];
  {a, u, v} (*a - НОД *)
];

{gcdEuclid, u, v} = extEuclid[f, g];
"GCD =" gcdEuclid
"u =" u
"v =" v

```

$$\text{Out[10]= } \text{GCD} = \left( \frac{6336}{49} - \frac{6336 x}{49} \right)$$

$$\text{Out[11]= } u = \left( -\frac{352 x^3}{175} - \frac{1936 x^2}{1225} + \frac{4752 x}{1225} + \frac{1584}{245} \right)$$

$$\text{Out[12]= } v = \left( -\frac{16 x^2}{35} + \frac{16 x}{49} + \frac{64}{49} \right)$$

*Нормировка полученных результатов*

```

In[13]:= coeff = CoefficientList[gcdEuclid, x]//Last

gcdEuclidN = (gcdEuclid / coeff)//ExpandAll
uN = (u / coeff)//ExpandAll
vN = (v / coeff)//ExpandAll

```

$$\text{Out[13]= } -\frac{6336}{49}$$

$$\text{Out[14]= } x - 1$$

$$\text{Out[15]= } \frac{7 x^3}{450} + \frac{11 x^2}{900} - \frac{3 x}{100} - \frac{1}{20}$$

$$\text{Out[16]= } \frac{7 x^2}{1980} - \frac{x}{396} - \frac{1}{99}$$

### Проверка

1. Сравним результат встроенной в язык функции и написанной нами.
2. Посмотрим на разложение Безу.
3. Сравним полученный результат с НОД.

```
In[17]:= SameQ[gcdAuto, gcdEuclidN]
bezout = (uN * f + vN * g)//ExpandAll
SameQ[bezout, gcdEuclidN]
```

```
Out[17]= True
```

```
Out[18]=  $x - 1$ 
```

```
Out[19]= True
```